# Maintainer: Felix Yan <felixonmars@archlinux.org>

pkgname=dhall-bash
pkgver=1.0.40
pkgrel=226
pkgdesc='Compile Dhall to Bash'
url='https://dhall-lang.org'
license=('BSD')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'dhall' 'haskell-neat-interpolation' 'haskell-shell-escape'
         'haskell-optparse-generic')
makedepends=('ghc' 'uusi')
source=("https://hackage.haskell.org/packages/archive/${pkgname}/${pkgver}/${pkgname}-${pkgver}.tar.gz")
sha512sums=('1936c795f843af2fd5556f0e2cb041218bdb3a205a0e35d215a24d235a615c2e6267f06cb4e58848c2dc6319bfbd306e3cae8321e41f25e9326f28737071b9a6')

prepare() {
    cd $pkgname-$pkgver
    uusi -u optparse-generic $pkgname.cabal
}

build() {
    cd $pkgname-$pkgver

    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname --enable-tests \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

check() {
    cd $pkgname-$pkgver
    runhaskell Setup test --show-details=direct
}

package() {
    cd $pkgname-$pkgver
    install -D -m744 register.sh   "${pkgdir}/usr/share/haskell/register/${pkgname}.sh"
    install -D -m744 unregister.sh "${pkgdir}/usr/share/haskell/unregister/${pkgname}.sh"
    runhaskell Setup copy --destdir="${pkgdir}"
    install -D -m644 LICENSE "$pkgdir"/usr/share/licenses/$pkgname/LICENSE
    rm -f "$pkgdir"/usr/share/doc/$pkgname/LICENSE
}
